
Split of f_EUTRA_IdleUpdated_ConnectToAdditionalPDN
	Reason for change
	Establishment of 2nd PDN connectivity needs to be implemented as parallel behaviour for C2K preamble.

	Summary of change
	New altsteps a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step1_2, a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3, a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step4 to allow implementation of default behaviour for establishment of 2nd PDN connectivity.

	TTCN module
	EUTRA_InitialRegistration.ttcn


Before
	  function f_EUTRA_IdleUpdated_ConnectToAdditionalPDN(EUTRA_CellId_Type p_CellId,

                                                      EUTRA_DelayForUserPlaneSignalling_Type p_WaitForIMS := waitForIMS) runs on EUTRA_PTC

  { /* preconditions:

       - connected mode: SRB1, SRB2 and one DRB

       - DRB: either DRB1 (if IMS is first PDN) or DRB8 (if IMS is second PDN)

       - IP routing is done for PDN_1 (as usual) and PDN_Internet (-> f_EUTRA_IdleUpdated_Step14_15)

       - PDN_Internet is mapped to PDN_2 i.e. uses second set of AddressInfo */

    var B3_Type v_PDNType := f_GetPdnType();           /* IPv4, IPv6 or IPv4v6 */

    var SRB_COMMON_IND v_ReceivedAsp;

    var NAS_MSG_Indication_Type v_NasInd;

    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;

    var ProcedureTransactionIdentifier v_PTId_UE;

    var template (omit) ProtocolConfigOptions v_Pco;

    var template (value) AccessPointName v_APN;

    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(secondPDN);        /* PDN_1 if IMS is second PDN, PDN_internet else */

    var EPS_BearerIdentity v_EpsBearerId := f_EUTRA_InitialRegistrationEpsBearerId(secondPDN);  /* tsc_EpsDefaultBearerId or tsc_EpsDefaultBearerId2ndPDN */

    var DRB_Identity v_DRB_Id := f_EUTRA_EpsBearerAssociatedDRB(v_EpsBearerId);             /* DRB1 or DRB8 */

    var PDN_Address v_PDN_Address;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(v_DRB_Id) };

    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_1;                                 /* non-IMS */

    if (tsc_EUTRA_Registration_PDN2isIMS) {                                                 /* for IMS @sic R5s141367 sic@ */

      v_BearerContextNumber := DEF_2;

    }

    // Step 1. The UE transmits a PDN CONNECTIVITY REQUEST message to request an additional PDN.

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNConnReqAdditionalPDN(v_PDNType,

                                                                                                                                -,

                                                                                                                                -,

                                                                                                                                cr_ESMInfoTransferFlag_Any ifpresent))))

      -> value v_ReceivedAsp;    /* NOTE: acc. to the prose the ESM information transfer flag is still "Not present or any allowed value"

                                  *       nevertheless if the UE sets the ESM information transfer flag to true this may cause issues

                                  *       as the generic procedure does not handle any requested ESM information transfer */

    v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

    v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

    v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

    v_APN := f_ESM_EvaluateAPN(p_CellId, secondPDN, v_PDNConnReq.accessPointName);    /* get APN without any ESM information transfer */

    v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    v_PDN_Address := f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas(v_Pco), v_PdnIndex);

    v_Pco := f_GetDefaultProtocolConfigOptions(v_Pco, v_PdnIndex);

    // Step 2. RRCConnectionReconfiguration message contains the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message.

    SRB.send(cas_SRB1_RrcNasPdu_REQ(p_CellId,

                                    cs_TimingInfo_Now,

                                    cs_RRCConnectionReconfiguration_DRB_Reconfig(tsc_RRC_TI_Def, v_DRB_ToAddModList),

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_508_ActivateDefEpsBearerContextRequest(v_EpsBearerId,

                                                                                             v_PTId_UE,

                                                                                             v_APN,

                                                                                             v_PDN_Address,

                                                                                             -,

                                                                                             v_Pco,

                                                                                             v_BearerContextNumber))));

    // Step 3. Receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND (p_CellId,

                                     cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 4. The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId))));

    // EXCEPTION: IP address allocation and/or IMS registration

    f_EUTRA_DelayForUserPlaneSignalling(p_WaitForIMS);   /* NOTE1: for multiple PDN if IMS is first PDN the IMS signalling will happen in parallel to the 2nd PDN connectivity procedure

                                                          *        => we cannot wait after the ATTCH_COMPLETE but need to wait here for the trigger of the IMS_PDN

                                                          * NOTE2: in case of IMS test cases in general p_WaitForIMS is dontWaitForIMS and the RRC connection release is controlled by the IMS test case */

  }


After
	  altstep a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step1_2(EUTRA_CellId_Type p_CellId,

                                                             EPS_BearerIdentity p_EpsBearerId) runs on EUTRA_PTC

  {

    var B3_Type v_PDNType := f_GetPdnType();           /* IPv4, IPv6 or IPv4v6 */

    var SRB_COMMON_IND v_ReceivedAsp;

    var NAS_MSG_Indication_Type v_NasInd;

    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;

    var ProcedureTransactionIdentifier v_PTId_UE;

    var template (omit) ProtocolConfigOptions v_Pco;

    var template (value) AccessPointName v_APN;

    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(secondPDN);        /* PDN_1 if IMS is second PDN, PDN_internet else */

    var DRB_Identity v_DRB_Id := f_EUTRA_EpsBearerAssociatedDRB(p_EpsBearerId);             /* DRB1 or DRB8 */

    var PDN_Address v_PDN_Address;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList := { cs_508_DRB_ToAddMod_DEFAULT_AM(v_DRB_Id) };

    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_1;                                 /* non-IMS */

    // Step 1. The UE transmits a PDN CONNECTIVITY REQUEST message to request an additional PDN.

    [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNConnReqAdditionalPDN(v_PDNType,

                                                                                                                                   -,

                                                                                                                                   -,

                                                                                                                                   cr_ESMInfoTransferFlag_Any ifpresent))))

      -> value v_ReceivedAsp    /* NOTE: acc. to the prose the ESM information transfer flag is still "Not present or any allowed value"

                                 *       nevertheless if the UE sets the ESM information transfer flag to true this may cause issues

                                 *       as the generic procedure does not handle any requested ESM information transfer */

      {

        v_NasInd := v_ReceivedAsp.Signalling.Nas[0];

        v_PDNConnReq := v_NasInd.Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

        v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

        v_APN := f_ESM_EvaluateAPN(p_CellId, secondPDN, v_PDNConnReq.accessPointName);    /* get APN without any ESM information transfer */

        v_Pco := v_PDNConnReq.protocolConfigurationOptions;

        v_PDN_Address := f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas(v_Pco), v_PdnIndex);

        v_Pco := f_GetDefaultProtocolConfigOptions(v_Pco, v_PdnIndex);

        if (tsc_EUTRA_Registration_PDN2isIMS) {                                                 /* for IMS @sic R5s141367 sic@ */

          v_BearerContextNumber := DEF_2;

        }

        // Step 2. RRCConnectionReconfiguration message contains the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message.

        SRB.send(cas_SRB1_RrcNasPdu_REQ(p_CellId,

                                        cs_TimingInfo_Now,

                                        cs_RRCConnectionReconfiguration_DRB_Reconfig(tsc_RRC_TI_Def, v_DRB_ToAddModList),

                                        cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                       cs_508_ActivateDefEpsBearerContextRequest(p_EpsBearerId,

                                                                                                 v_PTId_UE,

                                                                                                 v_APN,

                                                                                                 v_PDN_Address,

                                                                                                 -,

                                                                                                 v_Pco,

                                                                                                 v_BearerContextNumber))));

      }

  }

	  altstep a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    // Step 3. Receive RRCConnectionReconfigurationComplete

    [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId,

                                       cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)))

      {}

  }

	  altstep a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step4(EUTRA_CellId_Type p_CellId,

                                                           EPS_BearerIdentity p_EpsBearerId) runs on EUTRA_PTC

  {

    // Step 4. The UE transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message

    [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                         cr_508_ActivateDefEpsBearerContextAccept(p_EpsBearerId))))

      {}

  }

	  function f_EUTRA_IdleUpdated_ConnectToAdditionalPDN(EUTRA_CellId_Type p_CellId,

                                                      EUTRA_DelayForUserPlaneSignalling_Type p_WaitForIMS := waitForIMS) runs on EUTRA_PTC

  { /* preconditions:

       - connected mode: SRB1, SRB2 and one DRB

       - DRB: either DRB1 (if IMS is first PDN) or DRB8 (if IMS is second PDN)

       - IP routing is done for PDN_1 (as usual) and PDN_Internet (-> f_EUTRA_IdleUpdated_Step14_15)

       - PDN_Internet is mapped to PDN_2 i.e. uses second set of AddressInfo */

    var EPS_BearerIdentity v_EpsBearerId := f_EUTRA_InitialRegistrationEpsBearerId(secondPDN);  /* tsc_EpsDefaultBearerId or tsc_EpsDefaultBearerId2ndPDN */

    a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step1_2(p_CellId, v_EpsBearerId);

    a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(p_CellId);

    a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step4(p_CellId, v_EpsBearerId);

    // EXCEPTION: IP address allocation and/or IMS registration

    f_EUTRA_DelayForUserPlaneSignalling(p_WaitForIMS);   /* NOTE1: for multiple PDN if IMS is first PDN the IMS signalling will happen in parallel to the 2nd PDN connectivity procedure

                                                          *        => we cannot wait after the ATTCH_COMPLETE but need to wait here for the trigger of the IMS_PDN

                                                          * NOTE2: in case of IMS test cases in general p_WaitForIMS is dontWaitForIMS and the RRC connection release is controlled by the IMS test case */

  }


New global variable EUTRA_4_5A_16_State_Type
	Reason for change
	To handle establishment of 2nd PDN connectivity in default behaviour state information needs to be stored in global variable (NOTE: acc. to the TTCN-3 core language inout parameters are not allowed for altsteps being used as default behaviour)

	Summary of change
	New type definition for new global variable and appropriate function for access to the global variable

	TTCN module
	EUTRA_Component.ttcn


	  type enumerated EUTRA_4_5A_16_State_Type {step1_2, step3, step4, finished};

	  type record EUTRA_Ctrl_Type {                                         /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, UTRAN) */

    default                             Default,                        /* default behaviour being activated on EUTRA */

    boolean                             TestBody,                       /* true:  => fail may be assigend

                                                                           false: => inconc is assigned instead of fail */

    EUTRA_VerdictCtrl_Type              VerdictCtrl,                    /* Control the assignment of FAIL/INCOCN in default behaviour */

    EUTRA_HarqErrorHandling_Type        HarqErrorHandling,              /* @sic R5-106675 change 1 sic@ */

    EUTRA_UseBigGrants_Type             GrantType,                      /* @sic R5s140676 Change 2 sic@ */

    EUTRA_4_5A_16_State_Type            EUTRA_4_5A_16_State             /* @sic R5-145787 - Implementation for C2K sic@ */

  };

	  function fl_EUTRA_Common_Init(InitParams_Local_Type p_Params) runs on EUTRA_PTC

  {

    vc_EUTRA_Global.Ctrl.Default := null;

    vc_EUTRA_Global.Ctrl.TestBody := false;

    vc_EUTRA_Global.Ctrl.VerdictCtrl := normal;

    vc_EUTRA_Global.Ctrl.HarqErrorHandling := ignore;     /* @sic R5-106675 change 1 sic@ */

    vc_EUTRA_Global.Ctrl.GrantType := USE_DEF_GRANTS;     /* @sic R5s140676 Change 2 sic@ */

    vc_EUTRA_Global.Ctrl.EUTRA_4_5A_16_State := step1_2;  /* @sic R5-145787 - Implementation for C2K sic@ */

    // Initialise security parameters

    vc_EUTRA_Global.Security := f_EUTRA_Security_Init();

    // Initialise Cell array

    if (ischosen(p_Params.NonNas)) {

      vc_EUTRA_Global.CellArray := f_EUTRA_CellArray_Init(p_Params.NonNas.SysinfoCombination);

    } else {

      vc_EUTRA_Global.CellArray := f_EUTRA_NAS_CellArray_Init(p_Params.Nas.CellsOnSamePLMN, p_Params.Nas.SysinfoCombination, p_Params.Nas.CellFreq);

    }

    //Initialise Mobile Info

    vc_EUTRA_Global.MobileInfo.Tmsi := '00000000'O;

    vc_EUTRA_Global.MobileInfo.P_Tmsi := '00000000'O;

    vc_EUTRA_Global.MobileInfo.AssignedPdnType := '000'B;

    vc_EUTRA_Global.MobileInfo. LocalEmergencyNumberList := omit; /* @sic R5s141315 sic@ */

    f_EUTRA_ActivateDefault(StandardDefault);

  }

	  function f_EUTRA_4_5A_16_State_Get() runs on EUTRA_PTC return EUTRA_4_5A_16_State_Type

  {

    var EUTRA_4_5A_16_State_Type v_State := vc_EUTRA_Global.Ctrl.EUTRA_4_5A_16_State;

    return v_State;

  }

	  function f_EUTRA_4_5A_16_State_Set(EUTRA_4_5A_16_State_Type p_State) runs on EUTRA_PTC

  {

    vc_EUTRA_Global.Ctrl.EUTRA_4_5A_16_State := p_State;

  }


New altstep and default behaviour
	Reason for change
	alteps needed to be used as default behaviour

	Summary of change
	new altsteps: a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_DefaultHandler, a_EUTRA_IdleUpdated_ConnectToAdditionalPDN

	TTCN module
	EUTRA_InitialRegistration.ttcn


	  altstep a_EUTRA_IdleUpdated_ConnectToAdditionalPDN(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  { /* preconditions:

       - connected mode: SRB1, SRB2 and one DRB

       - DRB: either DRB1 (if IMS is first PDN) or DRB8 (if IMS is second PDN)

       - IP routing is done for PDN_1 (as usual) and PDN_Internet (-> f_EUTRA_IdleUpdated_Step14_15)

       - PDN_Internet is mapped to PDN_2 i.e. uses second set of AddressInfo */

    var EPS_BearerIdentity v_EpsBearerId := f_EUTRA_InitialRegistrationEpsBearerId(secondPDN);  /* tsc_EpsDefaultBearerId or tsc_EpsDefaultBearerId2ndPDN */

    var EUTRA_4_5A_16_State_Type v_State := f_EUTRA_4_5A_16_State_Get();

    [v_State == step1_2] a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step1_2(p_CellId, v_EpsBearerId) { 

      f_EUTRA_4_5A_16_State_Set(step3);

    }

    [v_State == step3] a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(p_CellId) {

      f_EUTRA_4_5A_16_State_Set(step4);

    }

    [v_State == step4] a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step4(p_CellId, v_EpsBearerId) { 

      f_EUTRA_4_5A_16_State_Set(finished);

    }

  }

	  altstep a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_DefaultHandler(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    [] a_EUTRA_IdleUpdated_ConnectToAdditionalPDN(p_CellId) { repeat; }

  }


Changes for f_EUTRA_Preamble_CDMA2000_PreRegistration
	Reason for change
	Establishment of 2nd PDN connectivity needs to be implemented in f_EUTRA_Preamble_CDMA2000_PreRegistration

	Summary of change
	Establishment of 2nd PDN connectivity added to f_EUTRA_Preamble_CDMA2000_PreRegistration as default behaviour

	TTCN module
	EUTRA_CommonProcedures_C2K.ttcn


	  function f_EUTRA_Preamble_CDMA2000_PreRegistration(EUTRA_CellId_Type p_CellId,

                                                     IDLEUPDATED_STATE_Type p_State,

                                                     boolean p_Is_1XRTT := false,

                                                     template (omit) RAND_CDMA2000 p_RAND_CDMA2000 := omit,

                                                    //@sic R5-145971 sic@

                                                    //EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                                     PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE) runs on EUTRA_PTC

  { /* @sic R5s130347: local GuardTimer sic@ */

    var NAS_MSG_Indication_Type v_NAS_Ind;

    //@sic R5-145971 sic@

    var EUTRA_ATTACH_Type v_ForcedAttach := NORMAL ;

    var template (omit) EMM_Cause v_EMMCause := omit;

    var boolean v_SecondPdnToBeReleased := false;

    timer t_GuardTimer := tsc_GuardTimePreamble;

    var default v_Procedure_4_5A_16;

    t_GuardTimer.start;

    // First switch cell back on

    f_EUTRA_SetCellPower(p_CellId, tsc_ServingCellRS_EPRE);

    // Switch on UE

    f_EUTRA_SwitchOnUEandStartIP(p_CellId); // @sic R5s120065 sic@

    // Now continue with the procedure

    v_NAS_Ind := f_EUTRA_IdleUpdated_Step1_4(p_CellId, PREAMBLE, NORMAL); // @sic R5s110176 sic@

    if (p_Is_1XRTT) {  //@sic R5-145971 sic@

      v_ForcedAttach := EPS_Only; // Force attach for EPS only

      if (v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.epsAttachType.typeValue == tsc_EAttachType_Combined) {

        v_EMMCause :=cs_EMM_Cause_tv(tsc_Cause_CSDomainNotAvailable);

      }

    }

    //Send co-ordination message for the Preparing C2K SS  for pre-registration

    f_IRAT_SendCoOrd(CDMA2000, cms_IRAT_Trigger);

    f_EUTRA_IdleUpdated_Step5_15(p_CellId,

                                 p_State,

                                 v_ForcedAttach,

                                 v_NAS_Ind,

                                 p_PDN_TypeToBeUsed,

                                 -,

                                 -,

                                 -,

                                 -,

                                 -,

                                 -,

                                 v_EMMCause);

    // steps 16

    f_EUTRA_NAS_AttachComplete(p_CellId, tsc_EpsDefaultBearerId, noDelay); //@sic R5-121276 R5s120891 sic@

    if (tsc_EUTRA_Registration_MultiplePDN) {   /* @sic R5-145787 - Implementation for C2K sic@ */

      v_Procedure_4_5A_16 := activate(a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_DefaultHandler(p_CellId));

    }

    // steps 16A-16B

    if (p_Is_1XRTT) {

      f_CSFBParametersProc_1xRTT(p_CellId, p_RAND_CDMA2000);

    }

    // steps 17-39b[HRPD] or 18 [1xRTT]

    f_CDMA2000_PreRegistration (p_CellId);

    if (tsc_EUTRA_Registration_MultiplePDN) {   /* @sic R5-145787 - Implementation for C2K sic@ */

      deactivate(v_Procedure_4_5A_16);

      while (f_EUTRA_4_5A_16_State_Get() != finished) {     /* 2nd PDN connectivity has not been (fully) established yet */

        a_EUTRA_IdleUpdated_ConnectToAdditionalPDN(p_CellId);

      }

      v_SecondPdnToBeReleased := true;

    }

    f_EUTRA_DelayForUserPlaneSignalling(); //@sic R5s120891  sic@

    // steps 41[HRPD] or 19 [1xRTT]

    f_EUTRA_RRC_ConnectionRelease(p_CellId);

    if (v_SecondPdnToBeReleased) {    /* @sic R5-145787 - Implementation for C2K sic@ */

      f_EUTRA_IdleUpdated_ReleaseSecondPDN(p_CellId);

    }

    t_GuardTimer.stop;

  }


